Eigenvalues and eigenvectors in Maple

Maple has commands for calculating eigenvalues and eigenvectors of matrices. Because (as you have seen in class) this is a complicated subject, there are a few twists and turns in the Maple implementation, too. As is to be expected, Maple's commands for eigenvalues and eigenvectors reside in the "`linalg`" library, so to begin, you must execute:

```
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

The most direct way to calculate eigenvalues and eigenvectors in Maple is to use the commands `eigenvals` and `eigenvects`. Here is a simple example:

```
> A:=matrix(3,3,[0,1,0,1,0,-3,0,-1,0]);
```

$$A := \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & -3 \\ 0 & -1 & 0 \end{bmatrix}$$

```
> eigenvals(A);
```

$$0, 2, -2$$

So this matrix has three distinct real eigenvalues (three eigenvalues are to be expected for a 3-by-3 matrix, but there may be repetitions, some may be complex numbers, etc.. more on this below). The `eigenvects` command gives even more information:

```
> EV:=eigenvects(A);
```

$$EV := [-2, 1, \{[-1, 2, 1]\}], [0, 1, \{[3, 0, 1]\}], [2, 1, \{[-1, -2, 1]\}]$$

The output of `eigenvects` is a fairly complicated data structure -- it is a list of lists. On the largest scale, the output has three parts -- corresponding to three eigenvalues. In general, the output of eigenvects has a part for each different eigenvalue of the matrix (so if there are duplicate eigenvalues, that value only gets one entry in the list). Now let's look at the structure of an individual entry (note how square brackets can be used to select one of the items from the list):

```
> EV[2];
```

$$[0, 1, \{[3, 0, 1]\}]$$

This part of the output indicates that 0 is an eigenvalue of multiplicity 1 (in other words, 0 is a simple (non-duplicated) root of the characteristic polynomial of the matrix A, and that [3,0,1] is a basis for the space of eigenvectors corresponding to the eigenvalue 0. The other parts of the output of eigenvects are interpreted similarly.

More difficult cases:

There are lots of complications that can arise, some of them (eigenvalues can be complex, there can be duplicate eigenvalues) arising from the mathematics itself and others (finding the eigenvalues can be difficult or impossible in "closed form", or the expressions for them can be complicated and confusing) arising from the use of the computer. We try to illustrate some of these here:

1. **Eigenvalues can be complex**: Here is an example of this:

```
> B:=matrix(3,3,[0,1,0,1,0,3,0,-1,0]);
```

$$B := \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 3 \\ 0 & -1 & 0 \end{bmatrix}$$

```
> eigenvals(B);
```

$$0, I\sqrt{2}, -I\sqrt{2}$$

Notice that the matrices A and B do not look very different -- but B two complex eigenvalues (complex eigenvalues of real matrices always occur in pairs of complex conjugates). But `eigenvects` still works:

```
> eigenvects(B);
```

$$[0, 1, \{[-3, 0, 1]\}], [I\sqrt{2}, 1, \{[-1, -I\sqrt{2}, 1]\}], [-I\sqrt{2}, 1, \{[-1, I\sqrt{2}, 1]\}]$$

2. **Eigenvalues can be repeated** : For example:

```
> C:=matrix(3,3,[4,1,1,6,3,2,-3,-1,0]);
```

$$C := \begin{bmatrix} 4 & 1 & 1 \\ 6 & 3 & 2 \\ -3 & -1 & 0 \end{bmatrix}$$

```
> eigenvals(C);
```

$$5, 1, 1$$

```
> eigenvects(C);
```

$$[5, 1, \{[-1, -2, 1]\}], [1, 2, \{[1, -3, 0], [0, -1, 1]\}]$$

The second part of this output indicates that 1 is an eigenvalue with multiplicity 2 -- and the two vectors given are two linearly independent eigenvectors corresponding to the eigenvalue 1. Any linear combination of these two vectors is also an eigenvector corresponding to the eigenvalue 1.

3. **Repeated eigenvalues need not have the same number of linearly independent eigenvectors attached** : For example:

```
> M:=matrix(3,3,[1,2,4,0,1,3,0,0,1]);
```

$$M := \begin{bmatrix} 1 & 2 & 4 \\ 0 & 1 & 3 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> eigenvals(M);
```

$$1, 1, 1$$

```
> eigenvects(M);
```

$$[1, 3, \{[1, 0, 0]\}]$$

Even though 1 is a root of the characteristic polynomial of M with multiplicity 3, there is only one linearly independent eigenvector corresponding to this eigenvalue. One concludes that M is not diagonalizable -- there is no way to find a basis for three-dimensional space that consists of eigenvectors of M.